

Maple ユーザは、新しい [MapleTA](#) パッケージによって、以前は Maple T.A.™ でのみ利用可能であった多くの学習用および評価用の内蔵コマンドにアクセスできるようになり、広範な問題を Maple で直接作成することができるようになりました。また、Maple T.A. の Maple で採点された問題を Maple に [インポート](#) することができます。Maple で作成した問題も Maple T.A. にエクスポートすることができます。

## ▼ 問題作成用と評価用の内蔵コマンド

Maple T.A.™ は、テストや課題を作成し、学生の回答と能力を自動的に評価する、使いやすいウェブベースのシステムです。複雑な自由形式の数式入力をサポートしており、回答をインテリジェントに評価でき、科学、テクノロジー、工学、数学 (STEM)、または数学を必要とするその他のコースに最適です。

Maple T.A. ソフトウェアでは Maple の計算エンジンを使用して学生の回答を評価します。数学領域では、同じ結果かを判断するのは難しい問題です。これを解決するには Maple のようなシンボリック言語を使用するツールが一番適切です。単純な問題でも (たとえば、求める解が  $x+y$  のときに学生の解が  $y+x$  であるような場合)、システムが学生の解の裏にある数学を理解できなければ難しい問題となります。

Maple T.A. は Maple のすべての数学操作にアクセスできます。また、Maple エンジンにアクセスしないアルゴリズムでも使用可能な、内蔵コマンドを保有しています。これを広範な問題に使用することができます。これらの内蔵コマンドの機能は、[MapleTA\[Builtin\]](#) サブパッケージに含まれています。

Maple T.A. の詳細は、[http://www.cybernet.co.jp/maple/product/maple\\_ta/](http://www.cybernet.co.jp/maple/product/maple_ta/) を参照してください。

## ▼ サンプル問題

たとえば、Maple T.A. の 2 項係数の概念を探るサンプル問題です。問題は、以下のとおりです。

*question* := "There are \$A balls of different colors in a sack.  
How many distinct combinations of \$B balls can you remove from the sack?":

ランダム変数 \$A および \$B を初期化するには、以下のようなコードを Maple T.A. のアルゴリズムセクションで目にするがあります。

```
$A:=range(5,10);$B:=range(1,4);$number:=binomial($A, $B);
```

Maple T.A. の [range](#) コマンドによって、指定された範囲内でランダム整数が生成されます。MapleTA:-Builtin パッケージを読み込むと、このコマンドが Maple 18 にでも使用できます。同じ作業を、[uses](#) コマンドを使用して、以下のプロシージャで陰的に行います。

```
initialize := proc( )  
  uses MapleTA:-Builtin;  
  local a, b, n;  
  a := range(5, 10);  
  b := range(1, 4);  
  n := binomial(a, b);  
  Grading:-Quiz:-Set( ` $A ` = a, ` $B ` = b, ` $number ` = n );  
end;
```

上記 [initialize](#) プロシージャの最終行によって、計算された変数と[採点](#)または表示に使われる変数の間の接続が作成されます。MapleTA[Builtin] パッケージでコマンドを使用して計算のみ行いたい場合は、この作業は不要です。

補足で、[Grading\[Quiz\]](#) コマンドを使用してこの問題を作成および採点する方法について説明します。

```
grade := proc( )  
  local n, r;  
  (n, r) := Grading:-Quiz:-Get( ` $number ` , ` $RESPONSE ` );  
  evalb(r=n);  
end;
```

問題を構成する基本要素が揃ったところで、Maple 上でこれを解くことや採点することができます。

```
Grading:-Quiz( question, grade, initialize );
```

There are  $10$  balls of different colors in a sack.

Check Answer

How many distinct combinations of  $1$  balls can you remove from the sack?

Try Another

**proc( ) local  $n, r$ ;  $n, r := \text{Grading:-Quiz:-Get}(\$number, \$RESPONSE)$ ;  $\text{evalb}(r=n)$  end proc** (2.1)

[Try Another] ボタンを数回クリックして、問題内の変数が変更することを確認してください。学生は、Maple 内で計算について調べることができます。[式] パレットに `choose` 記述があることを発見するかもしれません。袋に入っている 8 つのボールのうちの 2 つを取り出した場合、以下のように計算できます。

$$\binom{8}{2} \quad 28 \quad (2.2)$$

または、Maple の [binomial](#) コマンドを発見するかもしれません。

`binomial(8, 2);` 28 (2.3)

または、2 項係数の公式から解を得られるかもしれません。

$$\frac{8!}{2! \cdot (8-2)!} \quad 28 \quad (2.4)$$

学生が Maple を使用して解を得ようと調べることで、発見の機会を与えることができます。これにより、学生は、試験が終われば忘れてしまうというようなことなく、身につく方法で学習することができます。

最後に、Maple T.A. 内で直接使用できるように、この問題を[エクスポート](#)することができます。

```
MapleTA:-Export( Grading:-Quiz(question, grade, initialize, output = mapleta), "coursemodule.zip" );
```

Quiz コマンドを output=mapleta オプションで再実行し、エクスポートに最適な形式を作成しました (対話型の問題としてではなく)。この計算は「coursemodule.zip」ファイルを生成するために MapleTA[Export] に渡されます。このファイルは、Maple T.A. にコースモジュールとしてインポートすることができ、課題に使用できます。

同様に、Maple T.A. からエクスポートされたコースモジュールは Maple にインポートすることができ、これに含まれる問題を Grading パッケージの Quiz コマンドを使用して解くことや評価することができます。

## ▼ 問題を Maple T.A から (に) エクスポート (およびインポート)

Maple T.A. の「Maple で採点された」形式の問題を Maple にインポートして、Quiz コマンドで実行することができます。

MapleTA:-Import コマンドではコースモジュールの zip ファイルを読み込み、問題をコースモジュールマニフェストで XML に類似するレコードベースのデータ構造にすべてインポートします。サンプル構造は以下のように表示されることがあります (オプションフィールドは含まない)。

```
TAQuestion := Record("name" = " Quadratic Factor Question ", "mode" = "Maple", "text" = " Factor the following algebraic expression  $x^2 + Ax + B$ . Your answer should be in the form  $(x+a)(x+b)$ . ", "algorithm" = " $a:=rint(1,10);$b:=rint(2,10);$A:=int($a+$b);$B:=int($a*$b);$factor:=maple(/"factor(x^2+{A}*x+{B})/"); ", "mapleAnswer" = " $factor; ", "maple" = " evalb(($ANSWER)-($RESPONSE)=0); " );
```

インポートされた構造は Quiz コマンドによって識別され、Maple 内で適切な問題を構築します。

```
Grading:-Quiz( TAQuestion );
```

Factor the following algebraic expression  $x^2 + 14x + 48$ . Your answer should be in the form  $(x+a)(x+b)$ .

Check Answer

Try Another

コースモジュールには通常複数の問題が含まれています。

```
cmfile := cat(kernelopts(mapledir), "/data/help/MapleTA/coursemodule.zip") :
```

```
cmquestions := [ MapleTA:-Import( cmfile ) ] :
```

```
numelems(cmquestions);
```

18

(3.1

```
Grading:-Quiz( cmquestions[1] );
```

Convert  $614$  feet to meters. Answers require 4 significant digits. Do not include units in your response.

Check Answer


Try Another

```
Grading:-Quiz( cmquestions[3] );
```

式  $x^3 - x = 0$  を解きます。

Check Answer

Try Another



Enter only the expression for x, omitting "x =".

同様に、すべての Quiz ベースの問題は Maple T.A. に再度エクスポートすることができます。Quiz コマンドで 'output' = 'mapleta' オプションを設定すると、前述のとおり、レコード構造を生成します。これにより、このレコードを MapleTA:-Export コマンドに渡して、Maple T.A. にインポートするのに適したコースモジュールを生成することができます。

例として、問題を定義して見てみましょう。

```
qinit := proc()
  local p := nextprime( rand(2..7)() );
  Grading:-Quiz:-Set( `P` = p );
  [ p * rand(1..2)() + 1,
    p * rand(3..4)() + 1,
    p * rand(1..2)() + 0,
    p * rand(1..2)() + 2,
    p * rand(1..2)() - 1 ];
end;
```

Grading:-Quiz("For which values of X does ( X mod \$P) = 1", [1, 2], qinit, style = multipleselect) ;

For which values of X does ( X mod 5 ) = 1

Check Answer

Try Another

- 12
- 4
- 10
- 6
- 16

ここで、同じ問題パラメータを使用しますが、'output' オプションを指定してください。Maple T.A. で簡単に識別できるよう、問題に名前を付けます。

```
Q := Grading:-Quiz("For which values of X does ( X mod $P) = 1", [1, 2], qinit,  
style = multipleselect, output = mapleta, name = " Imported MS 15 (mod)")
```

```
Record(mode = "Multiple Selection", name = " Imported MS 15 (mod) ", algorithm = " $MAPLE_INIT = maple("randomize(): proc() local r := (proc  
( ) local p: p := nextprime(rand(2 .. 7)()): Grading:-Quiz:-Set(`\x24P` = p): [p*rand(1 .. 2)()+1, p*rand(3 .. 4)()+1, p*rand(1 .. 2)(), p*rand(1 .. 2)()  
+2, p*rand(1 .. 2)()-1] end proc)(): MapleTA:-ToString(op(r), Grading:-Quiz:-Get(`\x24P`),'print','sequence'): end(): ");  
$MCCHOICE1 = switch(0,$MAPLE_INIT);  
$MCCHOICE2 = switch(1,$MAPLE_INIT);  
$MCCHOICE3 = switch(2,$MAPLE_INIT);  
$MCCHOICE4 = switch(3,$MAPLE_INIT);  
$MCCHOICE5 = switch(4,$MAPLE_INIT);
```

```
$P = switch(5,$MAPLE_INIT); ", text = " For which values of X does ( X mod $P) = 1 ", type = "formula", answer = " 1, 2 ", _attributes  
= Record(...), choices = Record(...)
```

これをコースモジュールとしてエクスポートすることもできます。

```
if FileTools:-Exists("sample.zip") then  
  FileTools:-Remove("sample.zip");  
end if;
```

```
MapleTA:-Export( Q, "sample.zip" );
```

## ▼ 参照

[Maple 18 の新機能概要](#)、[MapleTA パッケージ](#)、[Maple T.A.積分](#)、[教育のための新しい機能](#)、[採点](#)